

**WCAPDF****PURPOSE**

Compute the standard wrapped-up Cauchy probability density function.

**DESCRIPTION**

The standard wrapped-up Cauchy distribution has the following probability density function:

$$f(x, p) = \frac{1 - p^2}{2\pi(1 + p^2 - 2p \cos(x))} \quad 0 \leq x < 2\pi \quad \text{(EQ Aux-329)}$$

where p is a shape parameter.

This distribution can be used as an alternative to the Von Mises distribution for symmetric, circular data.

**SYNTAX**

LET <y> = WCAPDF(<x>, <p>) <SUBSET/EXCEPT/FOR qualification>

where <x> is a number, parameter, or variable;

<p> is a number, parameter, or variable in the range (0,1) that specifies the shape parameter;

<y> is a variable or a parameter (depending on what <x> is) where the computed wrapped-up Cauchy pdf value is saved; and where the <SUBSET/EXCEPT/FOR qualification> is optional.

**EXAMPLES**

LET A = WCAPDF(3,0.5)

LET X2 = WCAPDF(X1,P)

**NOTE**

The general form of the wrapped-up Cauchy probability density function is:

$$f(x, p) = \frac{1 - p^2}{2\pi(1 + p^2 - 2p \cos(x - \mu))} \quad 0 \leq x < 2\pi \quad \text{(EQ Aux-330)}$$

where  $\mu$  is a location parameter.

**DEFAULT**

None

**SYNONYMS**

None

**RELATED COMMANDS**

WCACDF	=	Compute the wrapped-up Cauchy cumulative distribution function.
WCAPPF	=	Compute the wrapped-up Cauchy percent point function.
CAUCDF	=	Compute the Cauchy cumulative distribution function.
CAUPDF	=	Compute the Cauchy probability density function.
CAUPPF	=	Compute the Cauchy percent point function.
VONCDF	=	Compute the normal cumulative distribution function.
VONPDF	=	Compute the normal probability density function.
VONPPF	=	Compute the normal percent point function.

**REFERENCE**

“Continuous Univariate Distributions - Vol. 1,” 2nd. Ed., Johnson, Kotz, and Balakrishnan, John Wiley and Sons, 1994 (pp. 327-329).

**APPLICATIONS**

Circular Distributions

**IMPLEMENTATION DATE**

95/10

## PROGRAM

```

XILABEL X
YILABEL PROBABILITY
LET TWOPI = 2*PI
.
MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
LET P = 0.1
TITLE WRAPPED CAUCHY DISTRIBUTION - P = ^P
PLOT WCAPDF(X,P) FOR X = 0 0.01 TWOPI
LET P = 0.5
TITLE WRAPPED CAUCHY DISTRIBUTION - P = ^P
PLOT WCAPDF(X,P) FOR X = 0 0.01 TWOPI
LET P = 0.9
TITLE WRAPPED CAUCHY DISTRIBUTION - P = ^P
PLOT WCAPDF(X,P) FOR X = 0 0.01 TWOPI
LET P = 0.0
TITLE WRAPPED CAUCHY DISTRIBUTION - P = ^P
YLIMITS 0 0.5
PLOT WCAPDF(X,P) FOR X = 0 0.01 TWOPI
END OF MULTIPLY

```

